Pensieve header: Figuring out the ybax algebra by iterations. Continues ybax.nb.

ybax by iteration

Utilities

Canonical Form:

ybax

In[∘]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. 𝕖^x_ 𝕖^y_ :> 𝕖^(x+y) /. 𝕖^x_ :> 𝕖^CF[x]];
```

The Kronecker δ:

ybax

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality and multiplication of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

ybax

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

Zip and Bind

ybax

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_{i_})* := (u*)_i;
```

ybax

In[∘]:=
```
expand[sd_SeriesData] := MapAt[expand, sd, 3];
expand[ℰ_] := Expand[ℰ];
Zip_{}[P_] := P;
Zip_{ζ_,ζs___}[P_] := (expand[P // Zip_{ζs}] /. f_. ζ^d_. :> ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

ybax

In[∘]:=
```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^{-Q2} Zip_ζs[e^{Q1} (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

Upper to lower and lower to Upper:

ybax

In[∘]:=
```
U2l = {B_{i_}^{p_·} → e^{-p b_i}, B^{p_·} → e^{-p b}, T_{i_}^{p_·} → e^{p t_i}, T^{p_·} → e^{p t}, 𝒜_{i_}^{p_·} → e^{p α_i}, 𝒜^{p_·} → e^{p α}};
l2U = {e^{c_· b_i_ + d_·} :> B_i^{-c} e^d, e^{c_· b + d_·} :> B^{-c} e^d,
   e^{c_· t_i_ + d_·} :> T_i^c e^d, e^{c_· t + d_·} :> T^c e^d,
   e^{c_· α_i_ + d_·} :> 𝒜_i^c e^d, e^{c_· α + d_·} :> 𝒜^c e^d,
   e^{ℰ_} :> e^{Expand@ℰ}};
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^{Q}$ as a single"P". Here the z's are b and α and the ζ's are β and a.

ybax

In[∘]:=
```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_ζs[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U ];
LZip_{ζs_List} := LZip_{ζs,CF};
```

ybax

In[∘]:=
```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
   Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
   ] // LZip_{Flatten@Table[{β_{n@i}, τ_{n@i}, a_{n@i}}, {i, {is}}]} // QZip_{Flatten@Table[{ξ_{n@i}, y_{n@i}}, {i, {is}}]} ];
B_{L_List}[L_, R_] := Bind_L[L, R]; B_{is___}[L_, R_] := Bind_{is}[L, R];
```

The two halves

ybax

In[ ]:=
```
am_{i_,j_->k_} := 𝔼[(α_i + α_j) a_k, (ⅇ^{-α_j} ξ_i + ξ_j) x_k, 1 + O[ϵ]^2]
aΔ_{i_->j_,k_} := 𝔼[α_i (a_j + a_k), ξ_i (x_j + x_k), 1 + ϵ ξ_i x_k (-a_j + ξ_i x_j / 2) + O[ϵ]^2]
aS_{i_} := 𝔼[-α_i a_i, - ⅇ^{α_i} ξ_i x_i, 1 - ϵ ⅇ^{α_i} ξ_i x_i (a_i + ⅇ^{α_i} ξ_i x_i / 2) + O[ϵ]^2]
aSi_{i_} := 𝔼[-α_i a_i, -ⅇ^{α_i} ξ_i x_i, 1 - ϵ ⅇ^{α_i} ξ_i x_i (a_i - 1 + ⅇ^{α_i} ξ_i x_i / 2) + O[ϵ]^2]
```

ybax

In[ ]:=
```
bm_{i_,j_→k_} := 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, 1 - ϵ η_j y_k β_i + O[ϵ]^2]
bΔ_{i_->j_,k_} := 𝔼[β_i (b_j + b_k), η_i (ⅇ^{-b_k} y_j + y_k), 1 + ϵ η_i^2 y_j y_k ⅇ^{-b_k} / 2 + O[ϵ]^2]
bS_{i_} := 𝔼[-β_i b_i, -ⅇ^{b_i} η_i y_i, 1 - ϵ ⅇ^{b_i} η_i y_i (β_i + ⅇ^{b_i} η_i y_i / 2) + O[ϵ]^2]
bSi_{i_} := 𝔼[-β_i b_i, -ⅇ^{b_i} η_i y_i, 1 - ϵ ⅇ^{b_i} η_i y_i (β_i - 1 + ⅇ^{b_i} η_i y_i / 2) + O[ϵ]^2]
```

First check that on the generators this agrees with our conventions in the handout:

In[ ]:=
```
{{"[a,x]" → ((𝔼[0, 0, a_2 x_1] ~B_{1,2}~ am_{1,2→1}) 〚3〛 - (𝔼[0, 0, a_1 x_2] ~B_{1,2}~ am_{1,2→1}) 〚3〛),
   "[b,y]" → ((𝔼[0, 0, y_2 b_1] ~B_{1,2}~ bm_{1,2→1}) 〚3〛 - (𝔼[0, 0, y_1 b_2] ~B_{1,2}~ bm_{1,2→1}) 〚3〛)} /. z_1 → z,
 {"Δ[y]" → Last[𝔼[0, 0, y_1] ~B_1~ bΔ_{1→1,2}],
  "Δ[b]" → Last[𝔼[0, 0, b_1] ~B_1~ bΔ_{1→1,2}],
  "Δ[a]" → Last[𝔼[0, 0, a_1] ~B_1~ aΔ_{1→1,2}],
  "Δ[x]" → Last[𝔼[0, 0, x_1] ~B_1~ aΔ_{1→1,2}]},
 {
   "S(a)" → ((𝔼[0, 0, a_1] ~B_1~ aS_1) 〚3〛),
   "S(x)" → ((𝔼[0, 0, x_1] ~B_1~ aS_1) 〚3〛),
   "S(b)" → ((𝔼[0, 0, b_1] ~B_1~ bS_1) 〚3〛),
   "S(y)" → ((𝔼[0, 0, y_1] ~B_1~ bS_1) 〚3〛)
 } /. z_1 → z}
```

Out[ ]=
$\{\{[a,x] \to -x + O[\epsilon]^2,\ [b,y] \to -y\,\epsilon + O[\epsilon]^2\},\ \{\Delta[y] \to (B_2\, y_1 + y_2) + O[\epsilon]^2,$
$\Delta[b] \to (b_1 + b_2) + O[\epsilon]^2,\ \Delta[a] \to (a_1 + a_2) + O[\epsilon]^2,\ \Delta[x] \to (x_1 + x_2) - a_1\, x_2\, \epsilon + O[\epsilon]^2\},$
$\{S(a) \to -a + O[\epsilon]^2,\ S(x) \to -x - a\, x\, \epsilon + O[\epsilon]^2,\ S(b) \to -b + O[\epsilon]^2,\ S(y) \to -\frac{y}{B} + O[\epsilon]^2\}\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

In[ ]:=
```
{(aΔ_{1→1,2} ~B_2~ aΔ_{2→2,3}) ≡ (aΔ_{1→1,3} ~B_1~ aΔ_{1→1,2}), (bΔ_{1→1,2} ~B_2~ bΔ_{2→2,3}) ≡ (bΔ_{1→1,3} ~B_1~ bΔ_{1→1,2}),
 (am_{1,2→1} ~B_1~ am_{1,3→1}) ≡ (am_{2,3→2} ~B_2~ am_{1,2→1}), (bm_{1,2→1} ~B_1~ bm_{1,3→1}) ≡ (bm_{2,3→2} ~B_2~ bm_{1,2→1})}
```

Out[ ]= {True, True, True, True}

Δ is an algebra morphism

In[ ]:=
```
{am_{1,2→1} ~B_1~ aΔ_{1→1,2} ≡ (aΔ_{1→1,3} aΔ_{2→2,4}) ~B_{1,2,3,4}~ (am_{3,4→2} am_{1,2→1}),
 bm_{1,2→1} ~B_1~ bΔ_{1→1,2} ≡ (bΔ_{1→1,3} bΔ_{2→2,4}) ~B_{1,2,3,4}~ (bm_{3,4→2} bm_{1,2→1})}
```

Out[ ]= {True, True}

S is convolution inverse of id

In[∘]:=
```
{(aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{1,2→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{1,2→1}}
{(bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{1,2→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{1,2→1}}
```
Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2], \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2], \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

Si is the inverse of S

In[∘]:=
```
{aSi_1 ~ B_1 ~ aS_1 ≡ 𝔼[a_1 α_1, x_1 ξ_1, 1], aS_1 ~ B_1 ~ aSi_1 ≡ 𝔼[a_1 α_1, x_1 ξ_1, 1]}
{bSi_1 ~ B_1 ~ bS_1 ≡ 𝔼[b_1 β_1, y_1 η_1, 1], bS_1 ~ B_1 ~ bSi_1 ≡ 𝔼[b_1 β_1, y_1 η_1, 1]}
```
Out[∘]= {True, True}

Out[∘]= {True, True}

S is an algebra anti-(co)morphism

In[∘]:=
```
{am_{1,2→1} ~ B_1 ~ aS_1 ≡ (aS_1 aS_2) ~ B_{1,2} ~ am_{2,1→1}, bm_{1,2→1} ~ B_1 ~ bS_1 ≡ (bS_1 bS_2) ~ B_{1,2} ~ bm_{2,1→1}}
{aS_1 ~ B_1 ~ aΔ_{1→1,2} ≡ aΔ_{1→2,1} ~ B_{1,2} ~ (aS_1 aS_2), bS_1 ~ B_1 ~ bΔ_{1→1,2} ≡ bΔ_{1→2,1} ~ B_{1,2} ~ (bS_1 bS_2)}
```
Out[∘]= {True, True}

Out[∘]= {True, True}

Pairing

ybax

In[∘]:=
```
tP_{i_,j_} := 𝔼[β_i α_j, η_i ξ_j, 1 + ϵ η_i^2 ξ_j^2 / 4]
```

In[∘]:=
```
qfac[k_, q_] := (1 - q)^-k QPochhammer[q, q, k] // FunctionExpand
qfe[k_] := Normal[Series[qfac[k, E^ρ], {ρ, 0, 1}]] /. {ρ → ϵ}
Table[𝔼[0, 0, y_1^r b_1^s a_2^t x_2^u] ~ B_{1,2} ~ tP_{1,2} ≡ 𝔼[0, 0, Kδ_{r,u} Kδ_{s,t} qfe[r] s!],
   {r, 0, 4}, {s, 0, 4}, {t, 0, 4}, {u, 0, 4}] // Flatten // Union
```
Out[∘]= {True}

Pairing axioms

In[∘]:=
```
{(bm_{1,2→1} 𝔼[α_3 a_3, ξ_3 x_3, 1]) ~ B_{1,3} ~ tP_{1,3} ≡
   (𝔼[β_1 b_1, η_1 y_1, 1] 𝔼[β_2 b_2, η_2 y_2, 1] aΔ_{3→4,5}) ~ B_{1,4} ~ tP_{1,4} ~ B_{2,5} ~ tP_{2,5}
 , (bΔ_{1→1,2} 𝔼[α_3 a_3, ξ_3 x_3, 1] 𝔼[α_4 a_4, ξ_4 x_4, 1]) ~ B_{1,3} ~ tP_{1,3} ~ B_{2,4} ~ tP_{2,4} ≡
   (𝔼[β_1 b_1, η_1 y_1, 1] am_{3,4→3}) ~ B_{1,3} ~ tP_{1,3} }
```
Out[∘]= {True, True}

In[∘]:=
```
{(bS_1 𝔼[α_2 a_2, ξ_2 x_2, 1]) ~ B_{1,2} ~ tP_{1,2} ≡ (𝔼[β_1 b_1, η_1 y_1, 1] aS_2) ~ B_{1,2} ~ tP_{1,2},
 (bSi_1 𝔼[α_2 a_2, ξ_2 x_2, 1]) ~ B_{1,2} ~ tP_{1,2} ≡ (𝔼[β_1 b_1, η_1 y_1, 1] aSi_2) ~ B_{1,2} ~ tP_{1,2}}
```
Out[∘]= {True, True}

The Double

The double multiplication (should really bind the a’s and b’s separately) ybax

In[∘]:=

$$\text{Block}\left[\{i,j,k\},\ dm_{i\_,j\_\to k\_}=\left(\mathbb{E}[\beta_i b_i+\alpha_j a_j,\ \eta_i y_i+\xi_j x_j,\ 1]\ (a\Delta_{i\to h1,h2}\sim B_{h2}\sim a\Delta_{h2\to h2,h3})\ \left(b\Delta_{j\to t1,t2}\sim B_{t2}\sim b\Delta_{t2\to t2,t3}\right)\right)\sim B_{h3}\sim aSi_{h3}\sim B_{t1,h3}\sim\left(tP_{t1,h3}\right)\sim B_{t3,h1}\sim\left(tP_{t3,h1}\right)\sim B_{h2,j,i,t2}\sim\left(am_{h2,j\to k}\ bm_{i,t2\to k}\right)\right]$$

ybax

Out[∘]=

$$\mathbb{E}\Big[a_k\alpha_i+a_k\alpha_j+b_k\beta_i+b_k\beta_j,\ \frac{1}{\mathcal{A}_i\mathcal{A}_j}\left(y_k\mathcal{A}_i\mathcal{A}_j\eta_i+y_k\mathcal{A}_j\eta_j+x_k\mathcal{A}_i\xi_i+\mathcal{A}_i\mathcal{A}_j\eta_j\xi_i-B_k\mathcal{A}_i\mathcal{A}_j\eta_j\xi_i+x_k\mathcal{A}_i\mathcal{A}_j\xi_j\right),\ 1+\frac{1}{4\mathcal{A}_i\mathcal{A}_j}\big(-4y_k\mathcal{A}_j\beta_i\eta_j-4x_k\mathcal{A}_i\beta_j\xi_i+4x_k y_k\eta_j\xi_i+4a_kB_k\mathcal{A}_i\mathcal{A}_j\eta_j\xi_i+2y_k\mathcal{A}_j\eta_j^2\xi_i-6B_ky_k\mathcal{A}_j\eta_j^2\xi_i+2x_k\mathcal{A}_i\eta_j\xi_i^2-6B_kx_k\mathcal{A}_i\eta_j\xi_i^2+\mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2-4B_k\mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2+3B_k^2\mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2\big)\epsilon+O[\epsilon]^2\Big]$$

ybax

In[∘]:=

$$\text{Block}\left[\{i\},\ dS_{i\_}=\mathbb{E}[\beta_i b_1+\alpha_i a_2,\ \eta_i y_1+\xi_i x_2,\ 1]\sim B_{1,2}\sim\left(bSi_1\ aS_2\right)\sim B_{1,2}\sim dm_{2,1\to i}\right]$$

ybax

Out[∘]=

$$\mathbb{E}\Big[-a_i\alpha_i-b_i\beta_i,\ \frac{-y_i\mathcal{A}_i\eta_i-B_ix_i\mathcal{A}_i\xi_i+\mathcal{A}_i\eta_i\xi_i-B_i\mathcal{A}_i\eta_i\xi_i}{B_i},\ 1+\frac{1}{4B_i^2}\big(4B_iy_i\mathcal{A}_i\eta_i-4B_iy_i\mathcal{A}_i\beta_i\eta_i-2y_i^2\mathcal{A}_i^2\eta_i^2-4a_iB_i^2x_i\mathcal{A}_i\xi_i-4B_i^2x_i\mathcal{A}_i\beta_i\xi_i-4B_i\mathcal{A}_i\eta_i\xi_i+4a_iB_i\mathcal{A}_i\eta_i\xi_i+4B_i^2\mathcal{A}_i\eta_i\xi_i-4B_ix_iy_i\mathcal{A}_i^2\eta_i\xi_i+4B_i\mathcal{A}_i\beta_i\eta_i\xi_i-4B_i^2\mathcal{A}_i\beta_i\eta_i\xi_i+6y_i\mathcal{A}_i^2\eta_i^2\xi_i-2B_iy_i\mathcal{A}_i^2\eta_i^2\xi_i-2B_i^2x_i^2\mathcal{A}_i^2\xi_i^2+6B_ix_i\mathcal{A}_i^2\eta_i\xi_i^2-2B_i^2x_i\mathcal{A}_i^2\eta_i\xi_i^2-3\mathcal{A}_i^2\eta_i^2\xi_i^2+4B_i\mathcal{A}_i^2\eta_i^2\xi_i^2-B_i^2\mathcal{A}_i^2\eta_i^2\xi_i^2\big)\epsilon+O[\epsilon]^2\Big]$$

ybax

In[∘]:=

$$\text{Block}\left[\{i,j,k\},\ d\Delta_{i\_\to j\_,k\_}=\left(b\Delta_{i\to 3,1}\ a\Delta_{i\to 2,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{3,4\to k}\ dm_{1,2\to j}\right)\right]$$

ybax

Out[∘]=

$$\mathbb{E}\Big[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ y_j\eta_i+B_jy_k\eta_i+x_j\xi_i+x_k\xi_i,\ 1+\frac{1}{2}\left(B_jy_jy_k\eta_i^2-2a_jx_k\xi_i+x_jx_k\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
{
   "[a,y]" → ((𝔼[0, 0, y₂ a₁] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ a₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧),
   "[b,x]" → ((𝔼[0, 0, x₂ b₁] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, x₁ b₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧),
   "xy-qyx" → ((𝔼[0, 0, x₁ y₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (1 + ϵ) (𝔼[0, 0, y₁ x₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧)
  } /. {z_1 → z} // Expand // Factor
{
  "Δ(a)" → ((𝔼[0, 0, a₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
  "Δ(x)" → ((𝔼[0, 0, x₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
  "Δ(b)" → ((𝔼[0, 0, b₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
  "Δ(y)" → ((𝔼[0, 0, y₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧)
 } // Simplify
{
   "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ dS₁)⟦3⟧),
   "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ dS₁)⟦3⟧),
   "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ dS₁)⟦3⟧),
   "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ dS₁)⟦3⟧)
  } /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,y] \to -y + O[\epsilon]^2,\ [b,x] \to x\,\epsilon + O[\epsilon]^2,\ xy\text{-}qyx \to (1 - B) + a\,B\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^2,\ \Delta(x) \to (x_1 + x_2) - a_1\,x_2\,\epsilon + O[\epsilon]^2,$
$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^2,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^2\}$

Out[∘]= $\{S(a) \to -a + O[\epsilon]^2,\ S(x) \to -x - a\,x\,\epsilon + O[\epsilon]^2,\ S(b) \to -b + O[\epsilon]^2,\ S(y) \to -\frac{y}{B} + \frac{y\,\epsilon}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on double

(co)-associativity

In[∘]:=
```
{(dΔ₁→₁,₂ ~B₂~ dΔ₂→₂,₃) ≡ (dΔ₁→₁,₃ ~B₁~ dΔ₁→₁,₂),
 (dm₁,₂→₁ ~B₁~ dm₁,₃→₁) ≡ (dm₂,₃→₂ ~B₂~ dm₁,₂->₁)}
```

Out[∘]= {True, True}

Δ is an algebra morphism

In[∘]:=
```
dm₁,₂→₁ ~B₁~ dΔ₁→₁,₂ ≡ (dΔ₁→₁,₃ dΔ₂→₂,₄) ~B₁,₂,₃,₄~ (dm₃,₄→₂ dm₁,₂→₁)
```

Out[∘]= True

S is convolution inverse of id

In[∘]:=
```
{(dΔ₁→₁,₂ ~B₁~ dS₁) ~B₁,₂~ dm₁,₂→₁, (dΔ₁→₁,₂ ~B₂~ dS₂) ~B₁,₂~ dm₁,₂→₁}
```

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

S is a (co)-algebra anti-morphism

In[∘]:=
```
{dm₁,₂→₁ ~B₁~ dS₁ ≡ (dS₁ dS₂) ~B₁,₂~ dm₂,₁→₁, dS₁ ~B₁~ dΔ₁→₁,₂ ≡ dΔ₁→₂,₁ ~B₁,₂~ (dS₁ dS₂)} // Expand
```

Out[∘]= {True, True}

R-matrix

In[∘]:= $\mathbb{e}_{q\_,k\_}[x\_] := \mathbb{e}^\wedge\left(\sum_{j=1}^{k+1} \frac{(1-q)^j x^j}{j\,(1-q^j)}\right)$

ybax

In[∘]:= $R_{i\_,j\_} := \mathbb{E}\left[b_i a_j,\ y_i x_j,\ 1 - \epsilon\, y_i^2 x_j^2 / 4 + O[\epsilon]^2\right]$

In[∘]:= `Series[`$\mathbb{e}_{q,1}$`[z] /. {z → `$y_i x_j$`, q → 1 + ρ}, {ρ, 0, 1}] /. {ρ → ϵ}`

Out[∘]= $\mathbb{e}^{x_j y_i} - \frac{1}{4}\left(\mathbb{e}^{x_j y_i} x_j^2 y_i^2\right) \epsilon + O[\epsilon]^2$

Quasi-triangular axiom 1:

In[∘]:= $R_{1,2} \sim B_1 \sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4} R_{3,2}) \sim B_{2,4} \sim dm_{2,4\to 2}$

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:= $\left(\left(d\Delta_{1\text{->}1,2} R_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)\right) \equiv$
$\left(\left(d\Delta_{1\to 2,1} R_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{3,1\to 1}\, dm_{4,2\to 2}\right)\right)$

Out[∘]= True

Reidemeister 3:

In[∘]:= $\left((R_{1,2} R_{4,3} R_{5,6}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\right) \equiv$
$\left((R_{1,6} R_{2,3} R_{4,5}) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\right)$

Out[∘]= True

ybax

In[∘]:= $\text{Block}\left[\{i, j\},\ \overline{R}_{i\_,j\_} = \text{Expand} /@ R_{i,j} \sim B_j \sim dS_j\right]$

ybax

Out[∘]= $\mathbb{E}\left[-a_j b_i,\ -\frac{x_j y_i}{B_i},\ 1 + \frac{\left(-4 a_j B_i x_j y_i - 3 x_j^2 y_i^2\right)\epsilon}{4 B_i^2} + O[\epsilon]^2\right]$

Reidemeister 2

In[∘]:= $\left\{\left(\overline{R}_{1,2} R_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right),\ \left(R_{1,2} \overline{R}_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)\right\}$

Out[∘]= $\left\{\mathbb{E}\left[0, 0, 1 + O[\epsilon]^2\right],\ \mathbb{E}\left[0, 0, 1 + O[\epsilon]^2\right]\right\}$

Deriving the Drinfeld element u and its inverse ui

ybax

In[∘]:=
```
Block[{i}, {
  u_i_ = R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i},
  ui_i_ := R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→i}
 }]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[-a_i b_i,\ -\frac{x_i y_i}{B_i},\ B_i + \frac{\left(-4 a_i B_i^2 - 4 B_i x_i y_i - 4 a_i B_i x_i y_i - 3 x_i^2 y_i^2\right)\epsilon}{4 B_i} + O[\epsilon]^2\right],\ \text{Null}\right\}$

u and ui are inverses

In[∘]:= $(u_1\, ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\,0,\,1+O[\epsilon]^2\right]$

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}.
It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= $((u_1 \sim B_1 \sim dS_1)\; ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\,0,\,\frac{1}{B_1}+\frac{a_1\,\epsilon}{B_1}+O[\epsilon]^2\right]$

(* Needs fixing! *) So in our case S(u) = u z so S(u)u = $u^2\,z$ and v = u$z^{\frac{1}{2}}$ and finally C = uv^{-1} = $z^{-\frac{1}{2}} = e^{t_1/2}(1-\epsilon\, a_1)$.

ybax

In[∘]:=
```
Block[{i},
 {CC_i_ = 𝔼[0, 0, B_i^(1/2) ℯ^(-ϵ a_i/2) + O[ϵ]^2],
  CC̄_i_ = 𝔼[0, 0, B_i^(-1/2) ℯ^(ϵ a_i/2) + O[ϵ]^2]
 }]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[0,\,0,\,\sqrt{B_i}-\frac{1}{2}\left(a_i\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right],\,\mathbb{E}\left[0,\,0,\,\frac{1}{\sqrt{B_i}}+\frac{a_i\,\epsilon}{2\sqrt{B_i}}+O[\epsilon]^2\right]\right\}$

ybax

In[∘]:=
```
Block[{i, j}, {
   Kink_i_ = (R_{1,3} CC̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
   Kink̄_j_ = (R̄_{1,3} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j}
 }]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[a_i\,b_i,\,x_i\,y_i,\,\frac{1}{\sqrt{B_i}}+\frac{\left(2\,a_i-x_i^2\,y_i^2\right)\epsilon}{4\sqrt{B_i}}+O[\epsilon]^2\right],\right.$
$\left.\mathbb{E}\left[-a_j\,b_j,\,-\frac{x_j\,y_j}{B_j},\,\sqrt{B_j}+\frac{\left(-2\,a_j\,B_j^2-4\,a_j\,B_j\,x_j\,y_j-3\,x_j^2\,y_j^2\right)\epsilon}{4\,B_j^{3/2}}+O[\epsilon]^2\right]\right\}$

In[∘]:=
```
k2 = (R_{3,1} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i} /. ℯ → E;
k4 = (R̄_{3,1} CC̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j} /. ℯ → E;
Simplify@{Kink_i ≡ k2, Kink̄_j ≡ k4, (Kink_i Kink̄_j) ~ B_{i,j} ~ dm_{i,j→1}}
```

Out[∘]= $\{\text{True},\,\text{True},\,\mathbb{E}\left[0,\,0,\,1+O[\epsilon]^2\right]\}$

Reidemeister 2:

In[∘]:= $(R_{1,2}\,\bar{R}_{3,4}) \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{2,4} \sim dm_{2,4\to 2}$

Out[∘]= $\mathbb{E}\left[0,\,0,\,1+O[\epsilon]^2\right]$

Cyclic Reidemeister 2:

In[∘]:= $\left(R_{1,4}\,\overline{R}_{5,2}\,\overline{CC}_3\right) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \overline{CC}_1$

Out[∘]= True

Trefoil

ybax

In[∘]:= $Z = R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10};$
Do[$Z = Z \sim B_{1,r} \sim dm_{1,r\to 1}$, {r, 2, 10}];
Simplify /@ Z

ybax

Out[∘]= $$\mathbb{E}\left[0,\,0,\,\frac{B_1}{1-B_1+B_1^2}+\left(B_1\left(-B_1+2B_1^2+2B_1^4+a_1\left(-1+B_1-B_1^3+B_1^4\right)-2x_1y_1-B_1^3\left(3+2x_1y_1\right)\right)\epsilon\right)\Big/\left(1-B_1+B_1^2\right)^3+O[\epsilon]^2\right]$$

Timing[
$Z = R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10};$
Do[$Z = Z \sim B_{1,r} \sim dm_{1,r\to 1}$, {r, 2, 10}];
Simplify /@ Z]

---

ybax

In[∘]:= $b2t_{i\_} := \mathbb{E}\left[\alpha_i\,a_i - \beta_i\,t_i,\ \xi_i\,x_i + \eta_i\,y_i,\ 1 + \epsilon\,\beta_i\,a_i + O[\epsilon]^2\right]$
$t2b_{i\_} := \mathbb{E}\left[\alpha_i\,a_i - \tau_i\,b_i,\ \xi_i\,x_i + \eta_i\,y_i,\ 1 + \epsilon\,\tau_i\,a_i + O[\epsilon]^2\right]$

In[∘]:= $R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10}$

Out[∘]= $$\mathbb{E}\Bigg[a_5\,b_1+a_7\,b_3+a_2\,b_6-a_8\,b_8-a_9\,b_9-a_{10}\,b_{10},\ x_5\,y_1+x_7\,y_3+x_2\,y_6-\frac{x_8\,y_8}{B_8}-\frac{x_9\,y_9}{B_9}-\frac{x_{10}\,y_{10}}{B_{10}},\ \frac{\sqrt{B_8}\,\sqrt{B_9}\,\sqrt{B_{10}}}{\sqrt{B_4}}+\Bigg(\sqrt{B_{10}}\Bigg(\sqrt{B_9}\Bigg(\sqrt{B_8}\left(\frac{a_4}{2\sqrt{B_4}}-\frac{x_5^2\,y_1^2}{4\sqrt{B_4}}-\frac{x_7^2\,y_3^2}{4\sqrt{B_4}}-\frac{x_2^2\,y_6^2}{4\sqrt{B_4}}\right)+\frac{-2\,a_8\,B_8^2-4\,a_8\,B_8\,x_8\,y_8-3\,x_8^2\,y_8^2}{4\sqrt{B_4}\,B_8^{3/2}}\Bigg)+\frac{\sqrt{B_8}\left(-2\,a_9\,B_9^2-4\,a_9\,B_9\,x_9\,y_9-3\,x_9^2\,y_9^2\right)}{4\sqrt{B_4}\,B_9^{3/2}}\Bigg)+\frac{1}{4\sqrt{B_4}\,B_{10}^{3/2}}\sqrt{B_8}\,\sqrt{B_9}\left(-2\,a_{10}\,B_{10}^2-4\,a_{10}\,B_{10}\,x_{10}\,y_{10}-3\,x_{10}^2\,y_{10}^2\right)\Bigg)\epsilon+O[\epsilon]^2\Bigg]$$

In[∘]:= $\left(R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10}\right)\sim B_{Range[10]}\sim$ Product[$b2t_i$, {i, 10}]

Out[∘]= $\mathbb{E}\Big[-a_5\,t_1-a_7\,t_3-a_2\,t_6+a_8\,t_8+a_9\,t_9+a_{10}\,t_{10},\ \frac{1}{T_8\,T_9\,T_{10}}(T_8\,T_9\,T_{10}\,x_5\,y_1+T_8\,T_9\,T_{10}\,x_7\,y_3+T_8\,T_9\,T_{10}\,x_2\,y_6-T_9\,T_{10}\,x_8\,y_8-T_8\,T_{10}\,x_9\,y_9-T_8\,T_9\,x_{10}\,y_{10}),\ \frac{\sqrt{T_8}\,\sqrt{T_9}\,\sqrt{T_{10}}}{\sqrt{T_4}}+\frac{1}{4\sqrt{T_4}\,T_8^{3/2}\,T_9^{3/2}\,T_{10}^{3/2}}\big(4\,a_4\,T_8^2\,T_9^2\,T_{10}^2+4\,a_1\,a_5\,T_8^2\,T_9^2\,T_{10}^2+4\,a_2\,a_6\,T_8^2\,T_9^2\,T_{10}^2+4\,a_3\,a_7\,T_8^2\,T_9^2\,T_{10}^2-4\,a_8\,T_8^2\,T_9^2\,T_{10}^2-4\,a_8^2\,T_8^2\,T_9^2\,T_{10}^2-4\,a_9\,T_8^2\,T_9^2\,T_{10}^2-4\,a_9^2\,T_8^2\,T_9^2\,T_{10}^2-4\,a_{10}\,T_8^2\,T_9^2\,T_{10}^2-4\,a_{10}^2\,T_8^2\,T_9^2\,T_{10}^2-T_8^2\,T_9^2\,T_{10}^2\,x_5^2\,y_1^2-T_8^2\,T_9^2\,T_{10}^2\,x_7^2\,y_3^2-T_8^2\,T_9^2\,T_{10}^2\,x_2^2\,y_6^2-8\,a_8\,T_8\,T_9^2\,T_{10}^2\,x_8\,y_8-3\,T_9^2\,T_{10}^2\,x_8^2\,y_8^2-8\,a_9\,T_8^2\,T_9\,T_{10}^2\,x_9\,y_9-3\,T_8^2\,T_{10}^2\,x_9^2\,y_9^2-8\,a_{10}\,T_8^2\,T_9^2\,T_{10}\,x_{10}\,y_{10}-3\,T_8^2\,T_9^2\,x_{10}^2\,y_{10}^2\big)\,\epsilon+O[\epsilon]^2\Big]$

In[∘]:= Z = $\left(\left(\left(R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10}\right)\sim B_{Range[10]}\sim\text{Product}[b2t_i,\{i,10\}]\right)\ /.\ T\_\to T_1\right)\sim B_{Range[10]}\sim$ Product[$t2b_i$, {i, 10}]

Out[∘]= $\mathbb{E}\Big[a_5\,b_1+a_7\,b_3+a_2\,b_6-a_8\,b_8-a_9\,b_9-a_{10}\,b_{10},\ \frac{1}{B_1}(B_1\,x_5\,y_1+B_1\,x_7\,y_3+B_1\,x_2\,y_6-x_8\,y_8-x_9\,y_9-x_{10}\,y_{10}),\ B_1+\frac{1}{4\,B_1}\big(4\,a_1\,B_1^2+4\,a_4\,B_1^2-4\,a_8\,B_1^2-4\,a_9\,B_1^2-4\,a_{10}\,B_1^2-B_1^2\,x_5^2\,y_1^2-B_1^2\,x_7^2\,y_3^2-B_1^2\,x_2^2\,y_6^2+4\,a_1\,B_1\,x_8\,y_8-8\,a_8\,B_1\,x_8\,y_8-3\,x_8^2\,y_8^2+4\,a_1\,B_1\,x_9\,y_9-8\,a_9\,B_1\,x_9\,y_9-3\,x_9^2\,y_9^2+4\,a_1\,B_1\,x_{10}\,y_{10}-8\,a_{10}\,B_1\,x_{10}\,y_{10}-3\,x_{10}^2\,y_{10}^2\big)\,\epsilon+O[\epsilon]^2\Big]$

```
Timing[
 Do[Z = Z~B1,r~dm1,r→1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]= $\Big\{5.39063,\ \frac{B_1}{1-B_1+B_1^2}+\big(B_1\,\big(-B_1+2\,B_1^2+2\,B_1^4+a_1\,\big(-1+B_1-B_1^3+B_1^4\big)-2\,x_1\,y_1-B_1^3\,\big(3+2\,x_1\,y_1\big)\big)\,\epsilon\big)\Big/\big(1-B_1+B_1^2\big)^3+O[\epsilon]^2\Big\}$

In[∘]:=
```
Timing[
 Z = R1,5 R6,2 R3,7 CC4 Kink8 Kink9 Kink10 /. B_ → B1;
 Do[Print["doing ", r]; Z = Z~B1,r~dm1,r→1 /. B_ → B1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

```
doing 2
doing 3
doing 4
doing 5
doing 6
doing 7
doing 8
doing 9
doing 10
```

Out[ ]=

$$\left\{5.3125,\ \frac{B_1}{1-B_1+B_1^2}+\frac{1}{2 B_1 \left(1-B_1+B_1^2\right)^3}\left(2 a_1 B_1^2 \left(-1+B_1-B_1^3+B_1^4\right)-6 x_1^2 y_1^2+4 B_1^7 x_1^2 y_1^2-2 B_1^8 x_1^2 y_1^2+B_1^2 x_1 y_1 (5-6 x_1 y_1)+3 B_1 x_1 y_1 \left(-1+2 x_1 y_1\right)+B_1^6 \left(3+3 x_1 y_1-6 x_1^2 y_1^2\right)-B_1^5 \left(4+13 x_1 y_1+2 x_1^2 y_1^2\right)+B_1^4 \left(2+15 x_1 y_1+4 x_1^2 y_1^2\right)-B_1^3 \left(1+15 x_1 y_1+6 x_1^2 y_1^2\right)\right) \epsilon+O[\epsilon]^2\right\}$$